

课程名称：EDA 技术综合设计

设计报告名称：设计八 数字滤波器

班级：通信 214

姓名：王峤宇

学号：214022

一、设计内容及原理

数字滤波器从实现结构上划分, 有 FIR 和 IIR 两种。FIR (Finite Impulse Response) 滤波器的特点是它的冲击响应是有限的, 它跟过去的信号无关, 所以在使用时容易实现, 速度快。本次任务为设计一个 FIR 滤波器。

基础任务

设计任务: 利用板卡资源, 设计一个 8 阶的 FIR 低通滤波器, 采样频率是 5KHz, 阻带截止频率是 1KHz. 具体频率可以板卡实际资源为参照进行调整。

提高内容

设计任务: 在基础任务的基础上, 设置任意阶 FIR 滤波器, 要求采用模块化设计思想设计。

拓展任务

任务要求: 滤波器的架构方法可采用串行、并行等不同方法来实现, 采用不同方法设计实现 FIR 滤波器。

创新任务

设计思路: 不满足仅仅在仿真中实现的 FIR 滤波器, 将搭建的 FIR 滤波器综合并部署到 FPGA 开发板上, 利用串口与 FPGA 通信, PC 机为 FPGA 传输待滤波数据, FPGA 将滤波输出结果回传到 FPGA 上, PC 通过 MATLAB 或其他语言对数据进行验证。

十六倍过采样原理: 标准 UART 的 RXD 前端有一个“1 到 0 跳变检测器”, 当其连续接受到 8 个 RXD 上的地电平时, 该检测器就认为 RXD 线出现了起始位, 进入接受数据状态. 在接受状态, 接受控制器对数据位 7,8,9 三个脉冲采样, 并遵从三中取二的原则确定最终值. 采用这一方法的根本目的还是为了增强抗干扰, 提高数据传送的可靠性, 采样信号总是在每个接受位的中间位置, 可以避开数据位两端的边沿失真, 也可以防止接受时钟频率和发送时钟频率不完全同步引起的误差.

十六倍过采样本身也是对 UART 这种异步通信的一个延迟补偿, 如果没有过采样的话, 发送端和接收端的时钟将会 0.1 个波特率时钟周期, 引入 16 倍过采样后, 检测到发送的起始位后, 实现时钟同步, 由于同步的过程由 16 倍过采样的时钟控制, 同步之后的时钟相差仅为 0.1/16 个波特率时钟周期。

十六倍过采样的示意图如图 1 所示。




图 1: 十六倍过采样示意图

二、设计过程

基础任务

并行设计, 使用 Xilinx 提供的 IP 核进行设计, 源文件如下:

Listing 1: 并行八阶 FIR 滤波器

```

1  /* 8阶并行FIR滤波器 */
2  module fir_parallel (
3      input rst,          // 复位, 低有效
4      input clk,          // 工作频率, 即采样频率
5      input en,           // 输入数据开始信号
6      input [11:0] xin,   // 输入混合频率的信号数据
7      output out_rdy,     /* 输出数据有效 */
8      output [28:0] yout  // 输出数据
9  );
10 /* 滤波器设计 */
11 parameter ORDER = 8;          /* 默认为N为偶数 */
12 parameter MULT_NUM = ORDER/2 + 1; /* 系数为奇数 */
13 parameter DATA_NUM = (ORDER+1); /* 系数为奇数 */
14
15 wire signed [11:0] coe[MULT_NUM-1:0];
16 assign coe[0] = 12'd60;
17 assign coe[1] = 12'd165;
18 assign coe[2] = 12'd307;
19 assign coe[3] = 12'd432;
20 assign coe[4] = 12'd482;
21
22 // (0) 输入数据有效缓存, 保证输入数据延迟5个时钟时输出
23 reg [4:0] en_r;
24 always @(posedge clk or posedge rst) begin
25     if (rst) begin
26         en_r[4:0] <= 'b0;
27     end
28     else begin
29         en_r[4:0] <= {en_r[3:0], en};
30     end
31 end
32
33 // (1) 读取数据

```

```

34 reg signed [11:0] xin_reg[DATA_NUM-1:0];
35 integer i, j;
36 always @(posedge clk or posedge rst) begin
37     if (rst) begin
38         for (i = 0; i < DATA_NUM; i = i+1) begin
39             xin_reg[i] <= 12'b0;
40         end
41     end
42     else if (en) begin
43         xin_reg[0] <= xin;
44         for (j = 0; j < DATA_NUM - 1; j = j+1) begin
45             xin_reg[j+1] <= xin_reg[j]; // 周期性移位操作
46         end
47     end
48 end
49
50 // (2) 首尾相加，实现线性相位结构，延迟两个时钟
51 wire signed [12:0] add_reg[MULT_NUM-1:0];
52 assign add_reg[MULT_NUM-1] = {xin_reg[MULT_NUM-1][11], xin_reg[MULT_NUM-1]};
53 genvar l;
54 generate
55     for (l = 0; l < DATA_NUM/2; l = l+1) begin
56         c_addsub_0 your_instance_name (
57             .A(xin_reg[l]),           // 首部数据
58             .B(xin_reg[DATA_NUM-1-l]), // 尾部数据
59             .CLK(clk),               // 输入时钟
60             .S(add_reg[l])           // 输出数据
61         );
62     end
63 endgenerate
64
65 // (3) 乘法器，延迟3个时钟
66 wire signed [24:0] mout[MULT_NUM-1:0];
67 genvar k;
68 generate
69     for (k=0; k < MULT_NUM; k=k+1) begin
70         mult_gen_0 mult_parallel (
71             .CLK(clk),               // 时钟
72             .A(add_reg[k]),           // 13位的有符号数据
73             .B(coe[k]),               // 12位的有符号系数
74             .P(mout[k])               // 25位的输出数据
75         );
76     end
77 endgenerate
78
79 // (4) 积分累加，MULT_NUM组25bit数据
80 reg signed [28:0] sum;
81 always @(posedge clk or posedge rst) begin
82     if (rst) begin
83         sum <= 29'd0;
84     end
85     else begin

```

```

86         sum <= mout[0] + mout[1] + mout[2] + mout[3] + mout[4];
87     end
88 end
89
90 reg signed [28:0] yout_r;
91 reg out_rdy_r;
92 always @(posedge clk or posedge rst) begin
93     if (rst) begin
94         yout_r <= 29'd0;
95     end
96     else if (en_r[4]) begin
97         yout_r <= sum;
98         out_rdy_r <= 1'b1;
99     end
100    else begin
101        out_rdy_r <= 1'b0;
102    end
103 end
104 assign out_rdy = out_rdy_r;
105 assign yout = yout_r;
106 endmodule

```

通过 Matlab 程序生成用于仿真的数据结果如下:

Listing 2: 生成 12 位的波形数据

```

1 clear all;close all;clc;
2
3 fs = 5e3;
4 T = 1 / fs;
5 n = 0:2047;
6 t = n * T;
7 x = cos(2*pi*50*t) + 0.5*cos(2*pi*2000*t);
8 x = mapminmax(x);
9 x_dis = floor((2 ^ 11 - 1) * x);
10 fid = fopen('signal_data.txt', 'wt'); %写数据文件
11 for k=1:2048
12     B_si = dec2bin(x_dis(k) + (x_dis(k)<0)*2^12, 12);
13     fprintf(fid,'%s\n',B_si);
14 end
15 fprintf(fid,'');
16 fclose(fid);
17 stem(x(1:256));
18 title('原信号');

```

设计的滤波器如图 2所示, 采用最小二乘法设计, 阻带权值为 40, 通带为 1, 阻带截止频率为 1000Hz, 对输入为 200Hz+2000Hz 的信号进行滤波。




图 2: 滤波器设计




图 3: 加法器 IP

加法器 IP 核的设置如图 3所示, 使用 Xilinx 提供的 DSP Slice 资源上提供的乘法器和加法器实现。




图 4: 乘法器 IP 核配置 a




图 5: 乘法器 IP 核配置 b

乘法器的 IP 核配置结果如图 4和图 5所示, DSP Slice 资源提供的乘法器为流水线工作方式, 配置为 13 位和 12 位的符号数输入, 完成两个数据相加后和 FIR 系数的乘法运算, 在设置流水线阶段时, 设置为系统提示的最佳阶段, 三段流水线, 输出结果延迟三个时钟。仿真文件如下文件如下:

Listing 3: 8 阶 FIR 滤波器仿真文件

```
1 module fir_parallel_tb;
2     reg rst;
```

```

3    reg clk;
4    reg en;
5    reg [11:0] xin;
6    wire signed [28:0] yout;
7    wire out_rdy;
8    parameter SIN_DATA_NUM = 2048;
9    parameter FILE_PATH_A = "E:/Projects/matlabProject/fpgaTools/signal_data.txt";
10   parameter FILE_PATH_B = "E:/Projects/matlabProject/fpgaTools/signal_res.txt";
11   initial begin
12       rst = 0;
13       clk = 0;
14       #10;
15       rst = 1;
16       #10;
17       rst = 0;
18   end
19
20   reg [11:0] stimulus[0: SIN_DATA_NUM-1];
21   integer i, j;
22   integer file;
23   initial begin
24       $readmemb(FILE_PATH_A, stimulus);
25       file = $fopen(FILE_PATH_B, "w");
26       en = 1;
27       i = 0;
28       j = 0;
29       xin = 0;
30       #200;
31       forever begin
32           xin = stimulus[i];
33           @(negedge clk) begin
34               if (i == SIN_DATA_NUM-1) begin
35                   i = 0;
36               end
37               else begin
38                   i = i + 1;
39               end
40           end
41           @(posedge clk) begin
42               if (out_rdy) begin
43                   $fwrite(file, "%d\n", yout);
44                   j = j + 1;
45                   if (j == SIN_DATA_NUM) begin
46                       $finish;
47                       $fclose(file);
48                   end
49               end
50           end
51       end
52   end
53
54   fir_parallel fir_parallel_inst (

```

```

55     .rst(rst),
56     .clk(clk),
57     .en(en),
58     .xin(xin),
59     .yout(yout),
60     .out_rdy(out_rdy)
61 );
62
63     always #5   clk = ! clk ;
64
65 endmodule

```

提高任务

设计 15 阶的 FIR 滤波器, 采样频率 16kHz, 输入信号为语音信号, 使用低通滤波器保留 3.4kHz 以下的信号, 滤除 5000hz 单频噪声, 并且改善最终计算的求和过程。改善后的 FIR 滤波器源文件如下:

Listing 4: 15 阶并行 FIR 滤波器

```

1  /* 16 阶并行 FIR 滤波器 */
2  module fir_parallel_15 (
3      input  rst,           // 复位, 低有效
4      input  clk,           // 工作频率, 即采样频率
5      input  en,            // 输入数据开始信号
6      input  [11:0] xin,    // 输入混合频率的信号数据
7      output out_rdy,       // 输出数据有效
8      output [27:0] yout    // 输出数据
9  );
10 /* 滤波器设计 */
11 parameter ORDER = 15;           /* 默认 N 为奇数 */
12 parameter MULT_NUM = (ORDER+1)/2; /* 乘法器个数为偶数 */
13 parameter DATA_NUM = (ORDER+1); /* 数据个数为偶数 */
14
15 wire signed [11:0] coe[MULT_NUM-1:0];
16 assign coe[0] = -12'd94;
17 assign coe[1] = 12'd101;
18 assign coe[2] = 12'd190;
19 assign coe[3] = 12'd75;
20 assign coe[4] = -12'd153;
21 assign coe[5] = -12'd110;
22 assign coe[6] = 12'd362;
23 assign coe[7] = 12'd865;
24
25 // (0) 输入数据有效缓存, 保证输入数据延迟 5+6 个时钟时输出
26 reg [10:0] en_r;
27 always @(posedge clk or posedge rst) begin
28     if (rst) begin
29         en_r <= 'b0;
30     end
31     else begin
32         en_r <= {en_r[9:0], en};

```



```

33     end
34 end
35
36 // (1) 读取数据
37 reg signed [11:0] xin_reg[DATA_NUM-1:0];
38 integer i, j;
39 always @(posedge clk or posedge rst) begin
40     if (rst) begin
41         for (i = 0; i < DATA_NUM; i = i+1) begin
42             xin_reg[i] <= 12'b0;
43         end
44     end
45     else if (en) begin
46         xin_reg[0] <= xin;
47         for (j = 0; j < DATA_NUM - 1; j = j+1) begin
48             xin_reg[j+1] <= xin_reg[j]; // 周期性移位操作
49         end
50     end
51 end
52
53 // (2) 首尾相加，实现线性相位结构，延迟两个时钟
54 wire signed [12:0] add_reg[MULT_NUM-1:0];
55 genvar l;
56 generate
57     for (l = 0; l < DATA_NUM/2; l = l+1) begin
58         c_addsub_0 adder_12bit (
59             .A(xin_reg[l]),           // 首部数据
60             .B(xin_reg[DATA_NUM-1-l]), // 尾部数据
61             .CLK(clk),               // 输入时钟
62             .S(add_reg[l])           // 输出数据
63         );
64     end
65 endgenerate
66
67 // (3) 乘法器，延迟3个时钟
68 wire signed [24:0] mout[MULT_NUM-1:0];
69 genvar k;
70 generate
71     for (k=0; k < MULT_NUM; k=k+1) begin
72         mult_gen_0 mult_25bit (
73             .CLK(clk),           // 时钟
74             .A(add_reg[k]),       // 13位的有符号数据
75             .B(coe[k]),           // 12位的有符号系数
76             .P(mout[k])           // 25位的输出数据
77         );
78     end
79 endgenerate
80
81 // (4) 积分累加，8组25bit数据，输出28bit数据，使用三级加法器，延迟2*3个周期
82 wire signed [27:0] sum[6:0];
83 genvar m;
84 generate

```

```

85     for (m=0; m < 7; m=m+1) begin
86         if (m < 4) begin
87             c_addsub_1 adder_28bit (
88                 .A({3{mout[m*2][24]}}, mout[m*2]),
89                 .B({3{mout[m*2+1][24]}}, mout[m*2+1]),
90                 .CLK(clk),
91                 .S(sum[m])
92             );
93         end
94         else begin
95             c_addsub_1 adder_28bit (
96                 .A(sum[(m-4)*2]),
97                 .B(sum[(m-4)*2+1]),
98                 .CLK(clk),
99                 .S(sum[m])
100             );
101         end
102     end
103 endgenerate
104 reg signed [27:0] yout_r;
105 reg out_rdy_r;
106 always @(posedge clk or posedge rst) begin
107     if (rst) begin
108         yout_r <= 28'd0;
109     end
110     else if (en_r[10]) begin
111         yout_r <= sum[6];
112         out_rdy_r <= 1'b1;
113     end
114     else begin
115         out_rdy_r <= 1'b0;
116     end
117 end
118 assign out_rdy = out_rdy_r;
119 assign yout = yout_r;
120 endmodule

```

针对 15 阶的 FIR 滤波器的仿真文件如下:

Listing 5: 15 阶并行 FIR 滤波器的仿真文件

```

1 module fir_parallel_15_tb;
2     reg rst;
3     reg clk;
4     reg en;
5     reg [11:0] xin;
6     wire signed [28:0] yout;
7     wire out_rdy;
8     parameter SIN_DATA_NUM = 2048;
9     parameter FILE_PATH_A = "E:/Projects/matlabProject/fpgaTools/signal_data.txt";
10    parameter FILE_PATH_B = "E:/Projects/matlabProject/fpgaTools/signal_res.txt";
11    initial begin
12        rst = 0;

```

```

13     clk = 0;
14     #10;
15     rst = 1;
16     #10;
17     rst = 0;
18 end
19
20 reg [11:0] stimulus[0: SIN_DATA_NUM-1];
21 integer i, j;
22 integer file;
23 initial begin
24     $readmemb(FILE_PATH_A, stimulus);
25     file = $fopen(FILE_PATH_B, "w");
26     en = 1;
27     i = 0;
28     j = 0;
29     xin = 0;
30     #200;
31     forever begin
32         xin = stimulus[i];
33         @(negedge clk) begin
34             if (i == SIN_DATA_NUM-1) begin
35                 i = 0;
36             end
37             else begin
38                 i = i + 1;
39             end
40         end
41         @(posedge clk) begin
42             if (out_rdy) begin
43                 $fwrite(file, "%d\n", yout);
44                 j = j + 1;
45                 if (j == SIN_DATA_NUM) begin
46                     $finish;
47                     $fclose(file);
48                 end
49             end
50         end
51     end
52 end
53
54 fir_parallel_15 fir_parallel_inst (
55     .rst(rst),
56     .clk(clk),
57     .en(en),
58     .xin(xin),
59     .yout(yout),
60     .out_rdy(out_rdy)
61 );
62
63 always #5   clk = ! clk ;
64

```

65 `endmodule`

语音信号的 matlab 文件如下:

Listing 6: 语音信号文件处理

```
1  clear all; clc; close all;
2
3  [file, path] = uigetfile('*.wav'); % 读取音频
4  [xx, fs] = audioread([path, file]); % 读取音频
5
6  xx=xx-mean(xx); % 消除直流分量
7  x=xx/max(abs(xx)); % 幅值归一化
8  N=length(x); % 计算音频长度
9
10 time=(0:N-1)/fs; % 计算时间
11 noiseFreq=5000; % 单频噪声频率
12 noiseAmp=0.125; % 单频噪声幅度
13 signal=x + noiseAmp*cos(2*pi*noiseFreq*time); % 添加噪声
14 res = mapminmax(signal);
15 x_dis = floor((2 ^ 11 - 1) * res);
16 fid = fopen('signal_data.txt', 'wt'); % 写数据文件
17 for k=1:N
18     B_si = dec2bin(x_dis(k) + (x_dis(k)<0)*2^12, 12);
19     fprintf(fid, '%s\n', B_si);
20 end
21 fprintf(fid, ';');
22 fclose(fid);
23
24 % 作图
25 subplot 411; plot(time,x,'k'); grid;
26 title('原始语音'); ylabel('幅值'); xlabel('时间/s');
27
28 subplot 412; audioFFT = fft(x);
29 plot((0:round(N/2)-1) / N * fs, abs(audioFFT(1:round(N/2)) / N), 'k');grid;
30 title('原始语音频谱'); ylabel('幅值'); xlabel('freq(Hz)');
31
32 subplot 413; plot(time,signal,'k');grid;
33 title('带噪语音'); ylabel('幅值'); xlabel('时间/s');
34
35 subplot 414; audioFFT = fft(signal);
36 plot((0:round(N/2)-1) / N * fs, abs(audioFFT(1:round(N/2)) / N), 'k');grid;
37 title('带噪语音频谱'); ylabel('幅值'); xlabel('freq(Hz)');
```




图 6: 语音信号时域和频域图

语音信号频谱和时域图如图 6所示, 在 16000 采样速率的语音信号中添加了幅值为 0.125 单频噪声信号, 并将其转变为 12bit 的补码输出到 txt 文件中, 供 Verilog 仿真的系统级函数 \$readmemb 读取, 并输入到 FIR 滤波器之中。




图 7: 15 阶并行 FIR 滤波器综合结果

15 阶并行 FIR 滤波器的综合结果如图 7所示, 使用了 8 个乘法器和 8+7 个加法器, 极大的消耗资源, 但整个系统仅使用到一个 LUT。

拓展任务

15 阶串行 FIR 滤波器源文件如下:

Listing 7: 15 阶串行 FIR 滤波器源文件

```
1  `timescale 1ns / 1ns
2  /* 串行设计FIR滤波器 */s
3  module fir_serial (
4      input rst,          // 低电平复位
5      input clk,          // 系统工作时钟, 100MHz
6      input en,           // 输入数据有效信号
7      input [11:0] xin,   // 输入信号数据
8      output out_rdy,     // 输出数据有效信号
9      output [27:0] yout  // 输出数据
10 );
11 // 数据有效缓冲
12 reg [15:0] en_r;
13 always @ (posedge clk or posedge rst) begin
14     if (rst) begin
15         en_r[15:0] <= 'b0;
16     end
17     else begin
18         en_r[15:0] <= {en_r[14:0], en};
19     end
20 end
21
22 // 滤波器系数
23 wire signed [11:0] coe[7:0];
24 assign coe[0] = -12'd94;
25 assign coe[1] = 12'd101;
26 assign coe[2] = 12'd190;
27 assign coe[3] = 12'd75;
28 assign coe[4] = -12'd153;
29 assign coe[5] = -12'd110;
30 assign coe[6] = 12'd362;
31 assign coe[7] = 12'd865;
32
33 // (1) 输入数据移位部分
34 reg [2:0] cnt;
35 always @ (posedge clk or posedge rst) begin
36     if (rst) begin
37         cnt <= 3'b0;
38     end
39     else if (en || cnt != 0) begin
40         cnt <= cnt + 1'b1;    // 8个周期计数
41     end
42 end
43
44 // (1) 读入数据, 每8个周期读入一次数据, 延迟一个周期
45 integer i, j;
46 reg signed [11:0] xin_reg[15:0];
47 always @ (posedge clk or posedge rst) begin
```

```

48     if (rst) begin
49         for (i=0; i<16; i=i+1) begin
50             xin_reg[i] <= 12'b0;
51         end
52     end
53     else if (cnt == 3'd0 && en) begin // 每8个周期读入一次有效数据
54         xin_reg[0] <= xin;
55         for (j=0; j<15; j=j+1) begin
56             xin_reg[j+1] <= xin_reg[j]; // 数据移位
57         end
58     end
59 end

60
61 /* 3个周期 */
62 reg signed [11:0] add_a, add_b;
63 wire signed [12:0] add_s;
64 reg signed [11:0] coe_s[2:0];
65 wire [2:0] xin_index = cnt[2:0] >= 1 ? cnt[2:0] - 1 : 3'd7;
66 always @ (posedge clk or posedge rst) begin
67     if (rst) begin
68         add_a <= 12'b0;
69         add_b <= 12'b0;
70         coe_s[0] <= 12'b0;
71     end
72     else if (en_r[xin_index]) begin //from en_r[1]
73         add_a <= xin_reg[xin_index];
74         add_b <= xin_reg[15-xin_index];
75         coe_s[0] <= coe[xin_index];
76     end
77 end
78 always @(posedge clk or posedge rst) begin
79     if (rst) begin
80         coe_s[1] <= 12'b0;
81         coe_s[2] <= 12'b0;
82     end
83     else begin
84         coe_s[1] <= coe_s[0];
85         coe_s[2] <= coe_s[1];
86     end
87 end
88 c_addsub_0 adder (
89     .A(add_a),
90     .B(add_b),
91     .CLK(clk),
92     .S(add_s)
93 );
94
95 // (3) 乘法运算，使用一个流水线乘法器，3个周期
96 wire signed [24:0] mout;
97 /* 延迟3个时钟的流水线乘法器 */
98 mult_gen_0 mult (
99     .CLK(clk), // input wire CLK

```

```

100     .A(add_s), // input wire [12 : 0] A
101     .B(coe_s[2]), // input wire [11 : 0] B
102     .P(mout) // output wire [24 : 0] P
103 );
104
105 // (4) 积分累加, 延迟2个周期
106 wire signed [27:0] sum;
107 c_accum_0 your_instance_name (
108     .B(mout), // input wire [24 : 0] B
109     .CLK(clk), // input wire CLK
110     .BYPASS(cnt==3'd7), // input wire BYPASS
111     .Q(sum) // output wire [27 : 0] Q
112 );
113 reg out_rdy_r;
114 reg signed [27:0] yout_r; // 输出数据
115 always @(posedge clk) begin
116     if (en_r[15]) begin
117         out_rdy_r <= 1'b1;
118         yout_r <= sum;
119     end
120     else begin
121         out_rdy_r <= 1'b0;
122     end
123 end
124
125 assign yout = yout_r;
126 assign out_rdy = out_rdy_r;
127 endmodule

```

15 阶串行 FIR 滤波器的仿真文件如下:

Listing 8: 15 阶串行 FIR 滤波器的仿真文件

```

1  `timescale 1ns / 1ps
2
3  module fir_serial_tb;
4      //input
5      reg clk;
6      reg rst;
7      reg en ;
8      reg [11:0] xin ;
9      //output
10     wire signed [27:0] yout ;
11     wire out_rdy ;
12     parameter SIN_DATA_NUM = 10400;
13
14     initial begin
15         clk = 1'b0;
16         forever begin
17             #5 clk = ~clk;
18         end
19     end
20

```



```

21 parameter FILE_PATH_A = "E:/Projects/matlabProject/fpgaTools/extend/signal_data.txt";
22 parameter FILE_PATH_B = "E:/Projects/matlabProject/fpgaTools/extend/signal_res.txt";
23
24 initial begin
25     rst = 1'b1;
26     #30 rst = 1'b0;
27 end
28 reg [11:0] stimulus[0: SIN_DATA_NUM-1];
29 integer i, j, file;
30 initial begin
31     $readmemb(FILE_PATH_A, stimulus);
32     file = $fopen(FILE_PATH_B, "w");
33     en = 0;
34     i = 0;
35     j = 0;
36     xin = 0;
37     #200;
38     forever begin
39         repeat(7) @(negedge clk); //空置7个周期，每过16个周期给数据
40         en = 1;
41         xin = stimulus[i];
42         @(negedge clk);
43         en = 0; //输入数据有效信号只持续一个周期即可
44         if (i == SIN_DATA_NUM-1)
45             i = 0;
46         else
47             i = i + 1;
48     end
49 end
50 always @(posedge out_rdy) begin
51     $fwrite(file, "%d\n", yout);
52     j = j + 1;
53     if (j == SIN_DATA_NUM) begin
54         $finish;
55         $fclose(file);
56     end
57 end
58 fir_serial fir_filter_inst (
59     .rst(rst),
60     .clk(clk),
61     .en(en),
62     .xin(xin),
63     .out_rdy(out_rdy),
64     .yout(yout)
65 );
66 endmodule

```




图 8: 串行 FIR 综合结果

15 阶串行 FIR 滤波器的总结结果如图所示 8 所示, 仅使用了 1 个加法器、1 个乘法器以及一个累加器, 其数据输入要求间隔至少 8 个周期, 因为流水线以 8 个周期完成一次计算, 此外其数据相比于并行延迟较多, 同样的 DSP 单元, 延迟达到了 16 个周期。相比于串行设计用到了大量的查找表和触发器资源, 但也不到片上总资源的 0.5%。

创新设计

用到的 16 倍过采样串口通信模块的源文件如下:

Listing 9: 16 倍过采样 UART 模块

```

1  /* 时钟模块 */
2  module uart_clk_div (
3      input clk,
4      input rstn,
5      output clk_uart,
6      output clk_uart_16
7  );
8      // localparam step_freq = 32'd6597070;    /* 9600 * 16 */
9      localparam step_freq = 32'd79164837;    /* 115200 * 16 */
10
11     reg [31:0] cnt = 32'b0;
12     reg cnt_equal;
13     always@(posedge clk or negedge rstn) begin
14         if(!rstn)
15             cnt <= 32'b0;
16         else
17             cnt <= cnt + step_freq;
18     end
19     always @(posedge clk or negedge rstn) begin
20         if (!rstn) begin
21             cnt_equal <= 1'b0;
22         end else if (cnt < 32'h8000_0000) begin

```

```

23         cnt_equal <= 1'b0;
24     end else begin
25         cnt_equal <= 1'b1;
26     end
27 end
28 assign clk_uart_16 = cnt_equal;
29
30 /* 16倍分频 */
31 reg [3:0] cnt2 = 4'b0;
32 always @(posedge clk_uart_16 or negedge rstn) begin
33     if (!rstn) begin
34         cnt2 <= 4'b0;
35     end else begin
36         cnt2 <= cnt2 + 1'b1;
37     end
38 end
39 assign clk_uart = cnt2[3];
40 endmodule
41 /* 16倍过采样rx模块 */
42 module uart_rx(
43     input rxd,
44     input clk_uart_16,
45     input rst,
46     output rx_ne,
47     output reg [7:0] data_i
48 );
49 reg [7:0] data_i_buf;
50
51 // 接收器三个状态，格雷码
52 localparam st_IDLE = 3'b000; /* 空闲状态 */
53 localparam st_confirm = 3'b001; /* 确认接收 */
54 localparam st_RECEIVE = 3'b011; /* 接收数据 */
55 localparam st_STOP = 3'b010; /* 等待停止位 */
56 localparam st_RXNE = 3'b110; /* 接收非空 */
57 reg [2:0] rx_cur_st, rx_nxt_st;
58 reg [3:0] count1;
59 reg [3:0] count2;
60 /* 状态切换 */
61 always @(posedge clk_uart_16 or posedge rst) begin
62     if (rst) begin
63         rx_cur_st <= st_IDLE;
64     end
65     else begin
66         rx_cur_st <= rx_nxt_st;
67     end
68 end
69
70 /* 过采样时钟计数 */
71 always @(posedge clk_uart_16) begin
72     if (rx_cur_st == st_confirm || rx_cur_st == st_RECEIVE) begin
73         count2 <= count2 + 1'b1;
74     end else begin

```

```

75         count2 <= 4'b0000;
76     end
77 end
78
79 reg sampleBuf1;    /* 采样缓存单元 */
80 reg sampleBuf2;    /* 采样缓存单元 */
81 reg sampleBuf3;    /* 采样缓存单元 */
82 always @(posedge clk_uart_16 or posedge rst) begin
83     if (rst) begin
84         {sampleBuf1, sampleBuf2, sampleBuf3} <= 3'b0;
85     end
86     else if (rx_cur_st == st_confirm) begin
87         if (rx_d) begin
88             sampleBuf1 <= sampleBuf1 | 1'b1;
89         end else begin
90             sampleBuf1 <= sampleBuf1;
91         end
92     end else if (rx_cur_st == st_RECEIVE) begin
93         case (count2)
94             4'd7: sampleBuf1 <= rx_d;
95             4'd8: sampleBuf2 <= rx_d;
96             4'd9: begin
97                 casez ({sampleBuf1, sampleBuf2, rx_d})
98                     3'b11?, 3'b1?1, 3'b?11: begin
99                         sampleBuf3 <= 1'b1;
100                     end
101                     default: sampleBuf3 <= 1'b0;
102                 endcase
103             end
104             default: sampleBuf1 <= sampleBuf1;
105         endcase
106     end else begin
107         sampleBuf1 <= 1'b0;
108     end
109 end
110
111 always @(posedge clk_uart_16) begin
112     if (rx_cur_st == st_RECEIVE) begin
113         if (count2 == 4'd10) begin
114             count1 <= count1 + 1'b1;
115         end else begin
116             count1 <= count1;
117         end
118     end else begin
119         count1 <= 4'b0000;
120     end
121 end
122
123 always @(posedge clk_uart_16 or posedge rst) begin
124     if (rst) begin
125         data_i_buf <= 8'b0;
126     end

```

```

127     else if (rx_cur_st == st_RECEIVE) begin
128         if (count2 == 4'd10) begin
129             data_i_buf <= {sampleBuf3, data_i_buf[7:1]};
130         end else begin
131             data_i_buf <= data_i_buf;
132         end
133     end
134 end
135
136 /* 状态切换 */
137 always @(*) begin
138     case (rx_cur_st)
139         st_IDLE: begin
140             if (rx_d == 1'b0) begin /* 接收到开始信号 */
141                 rx_nxt_st <= st_confirm;
142             end else begin
143                 rx_nxt_st <= st_IDLE;
144             end
145         end
146         st_confirm: begin /* 确认起始信号 */
147             if (count2 == 4'b1000) begin
148                 if (!sampleBuf1) begin
149                     rx_nxt_st <= st_RECEIVE;
150                 end else begin
151                     rx_nxt_st <= st_IDLE;
152                 end
153             end else begin
154                 rx_nxt_st <= st_confirm;
155             end
156         end
157         st_RECEIVE: begin /* 接收到数据 */
158             if (count1 == 4'd9) begin
159                 rx_nxt_st <= st_STOP;
160             end else begin
161                 rx_nxt_st <= st_RECEIVE;
162             end
163         end
164         st_STOP: begin
165             if (rx_d) begin
166                 rx_nxt_st <= st_RXNE;
167             end else begin
168                 rx_nxt_st <= st_STOP;
169             end
170         end
171         st_RXNE: begin
172             if (rx_d == 1'b0) begin
173                 rx_nxt_st <= st_confirm;
174             end else begin
175                 rx_nxt_st <= st_IDLE;
176             end
177         end
178         default: rx_nxt_st <= st_IDLE;

```

```

179     endcase
180 end
181 always @(posedge rx_ne or posedge rst) begin
182     if (rst) begin
183         data_i <= 8'b0;
184     end
185     else begin
186         data_i <= data_i_buf;
187     end
188 end
189 /* 接收完成标志位 */
190 assign rx_ne = (rx_cur_st == st_RXNE)? 1'b1:1'b0;
191 endmodule
192 /* tx模块 */
193 module uart_tx(
194     input [7:0] data_o,
195     output reg txd,
196     input clk,
197     input rst,
198     input tx_en
199 );
200 // 发送机四个状态
201 localparam st_IDLE = 2'b00;    /* 空闲状态 */
202 localparam st_start = 2'b01;    /* 发送起始位 */
203 localparam st_data = 2'b10;    /* 发送数据 */
204 localparam st_end = 2'b11;    /* 发送结束 */
205
206 reg [2:0] count;
207 reg [1:0] tx_cur_st, tx_nxt_st;
208 reg [7:0] data_o_tmp;
209
210 reg tx_en_buf;
211 always @(posedge tx_en or posedge clk) begin
212     if (tx_en) begin
213         tx_en_buf <= 1'b1;
214     end else if (tx_cur_st == st_start) begin
215         tx_en_buf <= 1'b0;
216     end
217 end
218
219 always @(*) begin /* 组合逻辑 */
220     case (tx_cur_st)
221         st_IDLE: begin
222             if (tx_en_buf) begin
223                 tx_nxt_st <= st_start;
224             end else begin
225                 tx_nxt_st <= st_IDLE;
226             end
227         end
228         st_start: begin
229             tx_nxt_st <= st_data;
230         end

```

```

231     st_data: begin
232         if (count == 7) begin
233             tx_nxt_st <= st_end;
234         end else begin
235             tx_nxt_st <= st_data;
236         end
237     end
238     st_end: begin
239         if (tx_en_buf) begin
240             tx_nxt_st <= st_start;
241         end else begin
242             tx_nxt_st <= st_end;
243         end
244     end
245     default: tx_nxt_st <= st_IDLE;
246 endcase
247 end
248
249 always @(posedge clk) begin    /* 时序逻辑 */
250     if (rst) begin
251         tx_cur_st <= st_IDLE;
252     end else begin
253         tx_cur_st <= tx_nxt_st;
254     end
255 end
256
257 always @(posedge clk) begin
258     if (tx_cur_st == st_data) begin
259         count <= count + 1'b1;
260     end else begin
261         count <= 0;
262     end
263 end
264
265 always @(posedge clk) begin
266     if (tx_cur_st == st_start) begin
267         data_o_tmp <= data_o;
268     end else if (tx_cur_st == st_data) begin
269         data_o_tmp <= {1'b0, data_o_tmp[7:1]};
270     end
271 end
272
273 always @(posedge clk) begin
274     if (tx_cur_st == st_start) begin
275         txd <= 0;
276     end else if (tx_cur_st == st_data) begin
277         txd <= data_o_tmp[0];
278     end else begin
279         txd <= 1;
280     end
281 end
282 endmodule

```

基于串口通信的 FIR 硬件验证设计源文件如下:

Listing 10: 基于串口 FIR 硬件验证

```
1 module fir_top (
2     input clk,        /* 100MHz 时钟输入 */
3     input rxd,        /* rxd 接口 */
4     output txd,       /* txd 接口 */
5     input rst         /* 复位信号 */
6 );
7 // 时钟模块
8 wire clk_uart;
9 wire clk_uart_16;
10 uart_clk_div uart_clk_div (
11     .clk(clk),
12     .rstn(!rst),
13     .clk_uart(clk_uart),
14     .clk_uart_16(clk_uart_16)
15 );
16 // 串口接收模块
17 wire [7:0] data_rx;
18 wire rx_ne;
19 uart_rx uart_rx1 (
20     .clk_uart_16(clk_uart_16),
21     .rxd(rxd),
22     .rst(rst),
23     .data_i(data_rx),
24     .rx_ne(rx_ne)
25 );
26 // 串口发送模块
27 reg [7:0] data_tx;
28 reg tx_en;
29 uart_tx uart_tx1 (
30     .clk(clk_uart),
31     .txd(txd),
32     .rst(rst),
33     .data_o(data_tx),
34     .tx_en(tx_en)
35 );
36 wire out_rdy;
37 wire [27:0] yout;
38 reg [11:0] xin;
39 reg data_en;
40 fir_parallel_15 fir_parallel_15_inst (
41     .rst(rst),
42     .clk(clk_uart_16),
43     .en(data_en),
44     .xin(xin),
45     .out_rdy(out_rdy),
46     .yout(yout)
47 );
```



```

48
49  /* 一段式状态机 */
50  reg cnt_rx;
51  always @(posedge clk_uart_16 or posedge rst) begin
52      if (rst) begin
53          xin <= 12'b0;
54          cnt_rx <= 2'b0;
55          data_en <= 1'b0;
56      end
57      else if (cnt_rx == 1'b1) begin
58          if (rx_ne) begin
59              xin <= {xin[11 -: 4], data_rx};
60              cnt_rx <= 1'b0;
61              data_en <= 1'b1;
62          end
63      end
64      else if (rx_ne && data_rx[7:4] == 4'b1010) begin
65          xin <= {data_rx[3:0], 8'b0};
66          cnt_rx <= 1'b1;
67          data_en <= 1'b0;
68      end
69      else begin
70          cnt_rx <= 1'b0;
71          data_en <= 1'b0;
72      end
73  end
74
75  reg [9:0] cnt_tx; /* 2^5 + 2^4 */
76  reg cnt_tx_en;
77  reg [31:0] data_out_buf;
78  always @(posedge clk_uart_16 or posedge rst) begin
79      if (rst) begin
80          cnt_tx_en <= 1'b0;
81          cnt_tx <= 5'b0;
82          data_out_buf <= 32'b0;
83      end
84      else if (out_rdy) begin
85          data_out_buf <= {4'b1010, yout};
86          cnt_tx_en <= 1'b1;
87          cnt_tx <= 10'b1;
88      end
89      else if (cnt_tx == 10'b0) begin
90          cnt_tx_en <= 1'b0;
91      end
92      else if (cnt_tx_en == 1'b1) begin
93          if (cnt_tx[7:0] == 8'b1110_0000) begin
94              tx_en <= 1'b1;
95              data_tx <= data_out_buf[31 -: 8];
96              data_out_buf <= {data_out_buf[24:0], 8'b0};
97          end
98          else begin
99              tx_en <= 1'b0;

```

```

100         end
101         cnt_tx <= cnt_tx + 5'd1;
102     end
103 end
104 endmodule

```

基于串口通信的 FIR 硬件验证设计仿真文件如下:

Listing 11: 基于串口 FIR 硬件验证的仿真文件

```

1  module fir_top_tb;
2      reg clk;
3      wire rxd;
4      wire txd;
5      reg rst;
6      reg clk_uart;
7      reg clk_uart_16;
8      parameter SIN_DATA_NUM = 10400;
9      parameter FILE_PATH_A = "E:/Projects/matlabProject/fpgaTools/innovation/signal_data.txt";
10     parameter FILE_PATH_B = "E:/Projects/matlabProject/fpgaTools/innovation/signal_res.txt";
11     /* 复位 */
12     initial begin
13         rst = 1;
14         #30 rst = 0;
15     end
16     /* 时钟生成 */
17     initial begin
18         clk = 1'b0;
19         forever begin
20             #5 clk = ~clk;
21         end
22     end
23     initial begin
24         clk_uart_16 = 0;
25         forever begin
26             #271 clk_uart_16 = ~clk_uart_16;
27         end
28     end
29     initial begin
30         clk_uart = 0;
31         forever begin
32             #4336 clk_uart = ~clk_uart;
33         end
34     end
35
36     fir_top fir_top_inst (
37         .clk(clk),
38         .rxd(rxd),
39         .txd(txd),
40         .rst(rst)
41     );
42
43     reg tx_en;

```

```

44     reg [7:0] data_o;
45     uart_tx uart_tx (
46         .clk(clk_uart),
47         .txd(rxd),
48         .rst(rst),
49         .data_o(data_o),
50         .tx_en(tx_en)
51     );
52     wire rx_ne;
53     wire [7:0] data_i;
54     uart_rx uart_rx (
55         .rst(rst),
56         .clk_uart_16(clk_uart_16),
57         .rxd(txd),
58         .data_i(data_i),
59         .rx_ne(rx_ne)
60     );
61
62     reg [11:0] stimulus[0: SIN_DATA_NUM-1];
63     integer i;
64     initial begin
65         $readmemb(FILE_PATH_A, stimulus);
66         #100;
67         for (i = 0; i<SIN_DATA_NUM; i = i+1) begin
68             data_o <= {4'b1010, stimulus[i][11 -: 4]};
69             @(negedge clk_uart) tx_en <= 1'b1;
70             @(posedge clk_uart) tx_en <= 1'b0;
71             repeat(8)@(posedge clk_uart);
72             data_o <= {stimulus[i][7:0]};
73             @(negedge clk_uart) tx_en <= 1'b1;
74             @(posedge clk_uart) tx_en <= 1'b0;
75             repeat(4)@(posedge rx_ne);
76         end
77     end
78     reg signed [27:0] data_rx;
79     integer j;
80     integer k;
81     integer file;
82     initial begin
83         #100;
84         file = $fopen(FILE_PATH_B, "w");
85         for (j = 0; j<SIN_DATA_NUM; j = j + 1) begin
86             @(posedge rx_ne) data_rx[27:24] = data_i[3:0];
87             while (data_i[7:4] != 4'b1010) begin
88                 @(posedge rx_ne) data_rx[27:24] = data_i[3:0];
89             end
90             @(posedge rx_ne) data_rx[23:16] = data_i;
91             @(posedge rx_ne) data_rx[15:8] = data_i;
92             @(posedge rx_ne) data_rx[7:0] = data_i;
93             $fwrite(file, "%d\n", data_rx);
94         end
95     $finish;

```

```
96         $fclose(file);
97     end
98 endmodule
```

创新设计部分的约束文件如下:

Listing 12: 创新设计部分的约束文件

```
1 # 100MHz 时钟信号
2 set_property -dict {PACKAGE_PIN P17 IOSTANDARD LVCMOS33} [get_ports clk]
3 # 复位信号
4 set_property -dict {PACKAGE_PIN R11 IOSTANDARD LVCMOS33} [get_ports rst]
5 # 串口
6 set_property -dict {PACKAGE_PIN N5 IOSTANDARD LVCMOS33} [get_ports rxd]
7 set_property -dict {PACKAGE_PIN T4 IOSTANDARD LVCMOS33} [get_ports txd]
```

三、仿真结果

基础任务




图 9: 数据流过程仿真

针对输入数据计算流程的仿真结果如图 9所示, 当数据 en 有效时, 开始读入 xin 数据, 数据读入后可以看到, 其在延迟两个时钟后由加法器输出到对应的 add_reg 中, 然后数据开始不为 0, 乘法器开始工作, 从 add_reg 不为 0 开始, 延迟 3 时钟后乘法器开始输出, 再延迟一个时钟, 最后得到累加结果。在发生以上延迟后, 流水线进入填满状态, 每一个时钟输出一个有效数据。




图 10: 滤波结果仿真

滤波结果如图 10所示, 输入信号如下:

```
1 fs = 5e3;
2 T = 1 / fs;
3 n = 0:2047;
4 t = n * T;
5 x = cos(2*pi*50*t) + 0.5*cos(2*pi*2000*t);
```

将 50Hz 的信号视为目标信号, 2000Hz 的信号视为单频噪声, 可以看到通过 1k 截止频率的低通 FIR 滤波器后其信噪比明显得到改善, 2000Hz 信号被削弱。

提高任务




图 11: 滤波结果仿真




图 12: 音频信号理论情况

对输入信号, 即添加单频噪声的音频信号, 其时域图和频域图如图 12所示, 滤波结果仿真如图 11所示, 可以看到滤波结果的时域非常接近原始音频信号, 成功将音频信号中添加的单频噪声滤除。针对数据流的仿真如图 13所示, 从输入数据开始, 延迟两个时钟后, 得到 add_reg 中的结果, 之后延迟三个时钟后乘法器完成输出, 最后的累加积分的加法器也进行了优化, 因为使用了三级的延迟两个时钟的 DSP Slice 中的加法器单元, 累加求和延迟 6 个时钟, 整套计算系统为流水线模式, 经过 11 个时钟后, 流水线被填满之后, 每个时钟输出一个有效数据。




图 13: 数据流仿真




图 14: 仿真效果图

拓展任务

滤波效果仿真结果如图 14所示, 滤波效果理想, 可以看到加噪信号中的单频噪声被滤除。




图 15: 时序仿真图

串行设计的 FIR 滤波器时序仿真如图 15所示, 可以看到其与并行设计的最大差异也直接体现在仿真上, 不同的计算结果在并行 FIR 滤波器中体现为不同的数据, 不同的连线, 而在串行设计中只需要用到一个加法器, 一个乘法器和一个累加器, 计算过程中的 8 次加法通过时分复用同一个模块实现, 因此导致了较长的计算延迟, 使用 8 个时钟计算一次数据输出。图 15中可以看到, 不同的时钟周期, 模块的输出都不同, 而这些不同的数据属于一次计算, 最终比对计算结果, 可以看到其计算结果与并行设计的相同系数的 FIR 数值一样。

创新设计

创新设计部分的仿真结果如图 16所示, 可以看到数据通过串口发送到该模块后, 控制电路部分会等待有起始信号 1010 的数据, 同步输入数据, 每接收到两个数据有效数据后, 去除 1010 标识符后送入 FIR 滤波器, 当 FIR 滤波器输出时, 将数据缓存, 分 8 个 bit 通过串口的 tx 模块发送, 当上位机接收后, 再次传入数据, 之后计算完毕。




图 16: 创新设计的仿真结果

四、Matlab 验证结果

基础任务




图 17: 8 阶并行 FIR 滤波器滤波结果

通过 Matlab 计算比对滤波结果, 结果如图 17所示, 效果基本实现。

提高任务




图 18: Matlab 验证结果

通过 Matlab 将计算结果比对, 结果如图 18所示, 效果很好的实现, 但依然能够发现数值略微对不上, 推测为量化噪声引起, 可以看到信号未发生相位失真, 具备 FIR 特性。

拓展任务

15 阶滤波器对语音信号的 Matlab 滤波结果验证程序如下:

Listing 13: 语音信号滤波验证程序

```
1 [file, path] = uigetfile('*.wav'); % 读取音频
2 [xx, fs] = audioread([path, file]); % 读取音频
3
4 xx=xx-mean(xx); % 消除直流分量
5 x=xx/max(abs(xx)); % 幅值归一化
6 N=length(x); % 计算音频长度
7
8 time=(0:N-1)/fs; % 计算时间
9 noiseFreq=5000; % 单频噪声频率
10 noiseAmp=0.125; % 单频噪声幅度
11 signal=x + noiseAmp*cos(2*pi*noiseFreq*time); % 添加噪声
12 x_dis = floor((2 ^ 11 - 1) * mapminmax(signal)); % 引入量化噪声
13 b = [-94 101 190 75 -153 -110 362 865 865 362 -110 -153 75 190 101 -94];
14 x = mapminmax(x_dis);
15 x = filter(b, 1, x);
16
17 load('res.mat');
18 y = mapminmax(res);
```



```

19 % 作图
20 subplot 411; plot(time,x,'k'); grid;
21 title('Matlab 滤波信号时域'); ylabel('幅值'); xlabel('时间/s');
22
23 subplot 412; audioFFT = fft(x);
24 plot((0:round(N/2)-1) / N * fs, abs(audioFFT(1:round(N/2)) / N), 'k');grid;
25 title('Matlab 滤波信号频域'); ylabel('幅值'); xlabel('freq(Hz)');
26
27 subplot 413; plot(time,y,'k');grid;
28 title('FPGA 滤波时域'); ylabel('幅值'); xlabel('时间/s');
29
30 subplot 414; audioFFT = fft(y);
31 plot((0:round(N/2)-1) / N * fs, abs(audioFFT(1:round(N/2)) / N), 'k');grid;
32 title('FPGA 滤波频域'); ylabel('幅值'); xlabel('freq(Hz)');

```




图 19: 15 阶串行 FIR 滤波器验证结果

15 阶串行 FIR 的滤波结果与 15 阶并行的 FIR 滤波器滤波结果一致, 相比之下没有并行 FIR 滤波器输出的无效数据, 其值更加接近 Matlab 的滤波结果, 认为二者效果相同, 从图中也可以明显的观察到信号为发生相位失真, 具备 FIR 滤波器的特性。

五、创新内容的硬件验证结果




图 20: 基于串口的 FIR 滤波器设计验证实验结果

串口 FIR 硬件效果如图 20 所示, 输入数据为 12bit, 输出数据为 28bit, 串口使用 8bit 进行通信, 在输入数据和输出数据前添加 4bit 的前同步码, "1010", 即图 20 中串口调试助手显示的十六进制数据 0xA。PC 上位机向 FPGA 传输两个数据后, 输入数据达到 12bit, 进行一次计算, 输出 4 个 8bit 结果, 其中包含 28bit 数据。

针对创新部分的串口硬件验证过程, 设计 MATLAB 脚本程序, 通过串口实现 FIR 滤波, 其程序如下:

Listing 14: Matlab 通过串口与 FPGA 的 FIR 传输数据并验证

```
1 clear all;
2 hserial = serialport("COM14", 115200, "ByteOrder", "big-endian");
3 flush(hserial);
4
5 [file, path] = uigetfile('*.wav'); % 读取音频
6 [xx, fs] = audioread([path, file]); % 读取音频
7
8 xx = xx - mean(xx); % 消除直流分量
9 x = xx / max(abs(xx)); % 幅值归一化
10 N = length(x); % 计算音频长度
11 time = (0:N-1)/fs; % 计算时间
12 noiseFreq = 5000; % 单频噪声频率
13 noiseAmp = 0.125; % 单频噪声幅度
14 signal = x + noiseAmp * cos(2*pi*noiseFreq*time); % 添加噪声
15
```

```

16 q_12 = quantizer('Format', [12, 0]);
17 res = mapminmax(signal); % 归一化数据
18 x_dis_1 = floor((2 ^ 11 - 1) * res); % 数据量化
19 x_dis_2 = num2bin(q_12, x_dis_1);
20 x_dis_bin = strcat(repmat("1010", N, 1), x_dis_2); % 添加1010标识
21
22 q_8 = quantizer('Format', [8, 0], 'DataMode', 'ufixed');
23 q_32 = quantizer('Format', [32, 0]);
24 signal_filtered = zeros(1, N);
25 for index = 1:N
26     str = x_dis_bin(index);
27     data_out = bin2num(q_8, extractBefore(str, 9));
28     write(hserial, data_out, "uint8");
29     data_out = bin2num(q_8, extractAfter(str, 8));
30     write(hserial, data_out, "uint8");
31     data = read(hserial, 1, "uint32");
32     % 符号位扩展
33     for k = 0:3
34         data = bitset(data, 32-k, bitget(data, 28));
35     end
36     % 识别补码，转化为符号数输出
37     signal_filtered(index) = bin2num(q_32, dec2bin(data));
38 end
39 save result
40 % load result.mat
41 % 作图
42 xx = signal - mean(signal);
43 signal = xx/max(abs(xx));
44 xx = signal_filtered - mean(signal_filtered);
45 signal_filtered = xx/max(abs(xx));
46 subplot 611; plot(time,x,'k'); grid;
47 title('原始语音'); ylabel('幅值'); xlabel('时间/s');
48
49 subplot 612; audioFFT = fft(x);
50 plot((0:round(N/2)-1) / N * fs, abs(audioFFT(1:round(N/2)) / N), 'k');grid;
51 title('原始语音频谱'); ylabel('幅值'); xlabel('freq(Hz)');
52
53 subplot 613; plot(time,signal,'k');grid;
54 title('带噪语音'); ylabel('幅值'); xlabel('时间/s');
55
56 subplot 614; audioFFT = fft(signal);
57 plot((0:round(N/2)-1) / N * fs, abs(audioFFT(1:round(N/2)) / N), 'k');grid;
58 title('带噪语音频谱'); ylabel('幅值'); xlabel('freq(Hz)');
59
60 subplot 615; plot(time,signal_filtered,'k'); grid;
61 title('FPGA滤波后时域'); ylabel('幅值'); xlabel('时间/s');
62
63 subplot 616; audioFFT = fft(signal_filtered);
64 plot((0:round(N/2)-1) / N * fs, abs(audioFFT(1:round(N/2)) / N), 'k');grid;
65 title('FPGA滤波后频域'); ylabel('幅值'); xlabel('freq(Hz)');

```




图 21: FPGA 硬件验证结果

通过串口将待滤波数据传输到 FPGA 开发板上, 并通过串口读取其回传的滤波结果, 之后绘制其时域和频域图, 得到如图 21所示的结果, 改滤波器的效果显著, 且无线性相位失真发生, 效果明显, 证明了在提高任务和拓展任务中使用 xilinx 提供的 DSP48E 的 IP 核设计的 FIR 滤波器可以在硬件上有效的工作, 不仅仅局限于仿真结果。

六、问题解决

Vivado 的仿真过程中经常出现 UI 问题

解决: Vivado 的仿真界面虽然美观但是经常出现波形上数字消失, 而且模拟波形可以随意放缩比例尺, 不利于固定图像比例比对不同波形, 不利于观察波形, 最后简单的仿真过程, 如本次设计中的简单时序问题通过 Vivado 仿真, 但是复杂的如模拟波形数据, 流水线数据等通过 Modelsim 进行仿真。

Modelsim 和 Vivado 的联合仿真, 在修改文件后需要重启

解决: 通过查阅资料查看原理, 了解到 Vivado 在调用 Modelsim 仿真时是将文件交付 Modelsim 后, 其会对仿真文件和源文件进行编译, 然后再进行仿真, 并且编译结果是静态的, 仅仅重新编译还不够, 还需要手动更新。Vivado 交付给 Modelsim 的源文件保存在 Modelsim 中的 xil_defaultlib 库中, 更改文件后找到对应文件 update 以后再进行 recompile, 之后 reset 仿真, 再次仿真就是更改后的仿真结果。

串行 FIR 滤波器的设计过程无论如何改动, 结果都不正确

解决: 花费很长的时间观察仿真时序图, 起初未考虑到加法器的 2 周期时延与滤波器系数直接赋值无时延导致的不同步, 一直认为是累加器的累加过程中 BYPASS 信号控制不正确, 最后实在无法定位问题, 信号输入开始, 通过笔算的形式手动计算 FIR 结果和流程, 最终将问题定位到滤波器系数提

前加法器输出两个周期的问题,通过缓存移位寄存器人为滤波器系数的赋值延迟 2 个周期,解决问题,最终得到与并行 FIR 滤波器完全一致的计算结果。

七、心得体会

作为最后的创新设计实验,追求创新的前提下是比综合设计更加进一步复杂的设计过程。在本次创新设计中由于设计的是 FIR 滤波器,相比于乘法器或者求解最大公倍数等等简单的算法而言, FIR 滤波器的算法原理上不复杂,但随着阶数的上升其结构会极度的复杂。

如果直接在算法中使用“*”或者“+”会使得设计过程十分简单,但如果通过仿真的方式验证的话,仿真工具不会考虑乘法和加法得综合结果,直接以计算机语言得形式进行计算仿真,只要结构正确,就可以得到正确的结果,此外,直接使用 * 和 + 的综合结果是未知的,在未指定的情况下,甚至无法正确处理符号数

在实际的设计中,会使用并行或者串行加上流水线形式得乘法器和加法器进行设计。比如,对滤波器输出得最后累加积分过程中,并行设计的输出如果直接使用 7 个“+”将 8 个数据相加,综合工具甚至会将该累加过程综合成“一群”LUT 构成的组合逻辑,在实际应用中如此庞大得组合逻辑电路肯定无法完全避免竞争冒险等毛刺问题,以至于最终设计得到得 FIR 滤波器只能在很低的工作频率下运行,甚至有可能出错。由于创新设计得时间有限,采用 Xilinx FPGA 上集成的 DSP 模块中的乘法器和累加器实现 FIR 滤波器,实现设计安全和高效的均衡。

在设计阶段,花费了将近 3/4 的时间用于仿真,仿真能力得到了极大的提升,从最初的找不到仿真方向,到现在可以熟练的联合 Matlab 和 Modelsim 等仿真工具,熟悉了 Verilog 仿真文件中的若干系统级函数,也对 Modelsim 仿真工具更加熟练。

最后不满足于仅仅通过仿真实现 FIR 滤波器,尝试在硬件上验证所设计 FIR 滤波器的有效性,但局限于信号发生器、示波器等设备问题,放弃了 XADC 采样信号的并通过 DAC 再生的验证思路,着眼于使用串口将 FPGA 与上位机连接, FPGA 作为一个独立的计算单元供上位机调用,上位机通过 Matlab 程序验证 FPGA 输出的计算结果是否正确。在该创新内容的设计过程中,掌握了很多新的能力,如 Matlab 对系统串口资源的使用, Matlab 中补码的整数的转换等,以及更加复杂的仿真文件的编写等。

通过对数字滤波器和 Matlab 的设计,不仅加深了自己对于 FPGA 技术的理解,同时巩固了自己所学过的理论知识。